

Altium Designer – схемотехническое моделирование и типы моделей

(Продолжение)

Алексей Сабунин (Москва)

В предыдущей статье (см. СЭ № 6, 2009) было рассказано о процедуре выполнения моделирования в системе Altium Designer и форматах Spice-моделей, которые могут быть использованы для анализа. Также были рассмотрены встроенные модели и модели типа *.MDL. В данной статье завершается описание используемых моделей на примерах подсхем (subckt) и цифровых микросхем.

Модели, рассмотренные в предыдущей статье, описывают работу простых компонентов, таких как конденсатор, резистор, и т.д., вплоть до полевого транзистора и трансформатора. Для более сложных компонентов, таких как операционный усилитель, компаратор и др., используются модели более высокого уровня, которые представляют собой описание эквивалентной схемы данного устройства. Рассмотрим описание таких моделей на примере операционного усилителя.

Операционный усилитель (ОУ, OpAmp) представляет собой ИС, широко используемую в электронике. Реальная схема ОУ довольно сложна, и нет необходимости отражать все её свойства в каждом конкретном случае. В алгоритме Spice используются модели операционных усилителей трёх уровней:

- Level 1 – простейшая линейная модель, представляющая собой источник напряжения управляемый на-

пряжением (в дальнейшем – идеальный ОУ). Такой ОУ будет смоделирован для PSpice как усилитель с высоким входным сопротивлением, нулевым выходным сопротивлением и высоким коэффициентом усиления по напряжению;

- Level 2 – более сложная линейная модель, состоящая из двух каскадов и имитирующая два полюса передаточной функции ОУ, ограничение скорости нарастания входного напряжения, конечный коэффициент усиления и конечное выходное сопротивление;

- Level 3 – нелинейная модель. В ней учитываются ограничения на скорость нарастания входного напряжения, значения выходного сопротивления на постоянном и переменном токе, ток и напряжение смещения, запас по фазе на частоте единичного усиления, площадь усиления, коэффициент подавления

синфазного сигнала, реальные значения диапазона выходного напряжения и тока.

В учебных целях ознакомимся со структурой моделей первых двух типов. Идеальный ОУ и типичные значения параметров, составляющих его подсхемы, показаны на рисунке 1 (где $R_i = 1 \text{ ГОм}$, $A = 200\,000$ и $V(A3) = A \cdot (0 - V(A2))$). Эта модель служит для анализа на постоянном токе и низкой частоте. При необходимости можно изменять модель, учитывая другие свойства ОУ. На рисунке 1 показана схема проверки ОУ, а непосредственно подсхема ОУ показана внутри треугольника. Эта подсхема содержит резистор R_i и источник напряжения $E1$, управляемый напряжением на резисторе R_i . Резисторы, используемые на схеме, взяты из библиотеки *Miscellaneous Devices.IntLib*, которая находится в папке *Library* в директории установки *Altium Designer*. Кроме резисторов, на схеме использованы два источника, которые взяты из библиотеки *Simulation Sources.IntLib*, расположенной в папке *Library\Simulation* (описание используемых источников будет дано ниже).

На рисунке 1 показана схема включения ОУ с использованием отрицательной обратной связи по напряжению для PSpice. Резистор обратной свя-




Рис. 1. Идеальный операционный усилитель




Рис. 2. Настройки анализа узловых потенциалов

```
SOEL6
*SPICE Netlist generated by Advanced Sim server on 05.08.2009 11:30:51

*Schematic Netlist:
E1 A3 0 0 A2 200000
R1 A1 A2 1K
R2 A3 A2 10K
Ri A2 0 1G
VS A1 0 1V

.SAVE 0 A1 A2 A3 VS#branch @VS[z] @R1[i] @R2[i] @Ri[i] @E1[p] @R1[p] @R2[p]
.SAVE @Ri[p] @VS[p]

*PLOT OP -1 1 A=A1 A=R2 A=A3

.OPTION KeepLastSetup=False
*Selected Circuit Analyses:
.OP

.END
```

Рис. 3. Входной файл для анализа схемы




Рис. 4. Подключение модели к компоненту в библиотеке

зи R2 включен между выходом и инвертирующим входом, при этом неинвертирующий вход ОУ заземлён. Для проведения анализа необходимо указать обозначенные ранее узлы и выбрать вид анализа *Operating Point Analysis* (см. рис. 2). Напряжение на узле A3 равно -9,999 В, таким образом, коэффициент передачи близок к -10 В/В.

В рассмотренной схеме не был использован ОУ, а рассматривалась эквивалентная схема, представляющая идеальный ОУ. В результате моделирования был создан файл с расширением *.nscx, который показан на рисунке 3. Структура этого файла была описана в СЭ № 6, 2009, и строки, начинающиеся с E1 и Ri, относятся к эквивалентной схеме ОУ, тогда как остальная часть схемы необходима лишь для проверки работы ОУ.

Теперь можно описать собственно ОУ следующим образом:

```
.SUBCKT OPAMP A2 0 A3
E1 A3 0 0 A2 200000
Ri A2 0 1G
.ENDS
```

В первой строке после команды *.SUBCKT* (перед которой не должно быть пробела!) идёт название модели и перечисление выводов, далее – список элементов подсхемы. В нашем случае элемент E1 задаёт напряжение между узлами A3 и 0 в зависимости от напряжения между узлами A2 и 0. Цепь с названием 0 является опорной, относительно которой происходит измерение напряжений, и на схеме представлена названием GND.

Попробуем создать самостоятельный усилитель и добавить к нему модель для *PSpice*. Для начала создадим новый файл модели командой *File > New > Mixed-Signal Simulation > AdvancedSim Sub-Circuit*, вставим в него полученную ранее модель (выделенную полужир-

ным шрифтом) и сохраним под названием *OPAMP*. Далее создадим новую библиотеку и добавим в неё символ операционного усилителя (о создании библиотек и компонентов см. СЭ № 6, 2008). Теперь необходимо подключить модель к компоненту, для чего в списке моделей нажимаем кнопку *Add Simulation* (см. рис. 4), в списке *Model Sub-Kind* выбираем вариант *Spice Subcircuit*, что соответствует моделям в формате SKT, и после этого по кнопке *Browse* выбираем созданную ранее модель.

При подключении модели к компоненту следует помнить один важный момент: порядок перечисления выводов в первой строке модели соответствует порядку нумерации выводов на символе! В нашем случае этот порядок совпал, но это далеко не всегда так, и в особенности у компонентов, которые имеют несколько ячеек внутри одного корпуса (поскольку модель описывает лишь логику одной такой ячейки). Поэтому, находясь в окне *Sim Model – General/Spice Subcircuit*, следует перейти на вкладку *Port MAP* и при необходимости указать соответствие выводов символа выводам модели. Особое внимание следует обратить на структуру проекта (см. рис. 4). Здесь все используемые файлы находятся в рамках одного проекта *SOEL7* (схема, библиотека, модель). Поэтому при подключении модели к компоненту она была обнаружена в выпадающем списке, а при размещении компонента на схеме его библиотека находится в списке подключенных библиотек.

Рассмотрим случай, когда необходимо получить частотные характеристики операционного усилителя. Для этого следует использовать модель второго уровня, которая учитывает изменение параметров ОУ при изменении частоты. Для ОУ с типовыми ха-

рактеристиками предлагается модель, показанная на рисунке 5 [1].

Исследуем модель, которая включает $R_{in} = 1$ МОм, $R_O = 50$ Ом, $R_{i1} = 1$ кОм, $C = 15,92$ мкФ и EG с коэффициентом усиления по напряжению $A = 100$. Последний параметр представляет собой низкочастотный коэффициент усиления, или коэффициент усиления по постоянному току при разомкнутой обратной связи. При использовании этих значений, получим выходное напряжение на частоте $f_c = 10$ Гц, при которой выходное напряжение снижается на 3 дБ.

Для проведения частотного анализа, в настройках *Analyses Setup* выбираем вариант *AC Small Signal Analysis*, с параметрами *Start Frequency = 1*, *Stop Frequency = 1M*, *Sweep Type = Decade*, *Test Point = 100*. В результате моделирования будет получен график, показанный на рисунке 6, который по умолчанию имеет линейную шкалу. Для изменения типа шкалы на логарифмический необходимо выполнить двойной щелчок на оси OX (оси частоты) и в настройке *Grid Type* выбрать *Logarithmic* (см. рис. 6).

После отладки схемы её можно представить в виде самостоятельной модели, которая в данном случае примет следующий вид:

```
.SUBCKT OPAMP A2 0 A3
E2 X6 0 X4 0 1
EG X3 0 X2 X1 100
C1 X4 0 15.92u
Ri1 X3 X4 1K
Rin X2 X1 1MEG
RO X6 X5 50
.ENDS
```

Использование этой модели ничем не отличается от случая, описанного выше. Аналогичным образом можно создавать более сложные модели, которые будут содержать в своей структуре не толь-




Рис. 5. Использование подмодели OY для получения АЧХ

ко простейшие элементы (резисторы, конденсаторы и источники), но и более сложные, описываемые моделями в формате MDL (диоды, транзисторы).

В качестве примера прокомментируем модель компаратора LM339 (см. рис. 7), которая имеется в одной из библиотек программы и была обнаружена с помощью поиска. Здесь в заголовке модели несколько строк, начинающиеся с символа «*», являются комментарием. Далее в строке, которая начинается с команды .SUBCKT, указано название модели и перечислены её выводы. Затем идёт описание подсхемы: F, I, V – различные источники, и далее строка «Q1 9 21 7 QIN» говорит о том, что транзистор Q1 подключен к цепям 9, 21, 7 и для него используется модель QIN. После этого приведено текстовое описание модели QIN с помощью команды .MODEL.

Как видно из модели компаратора и из приведённых выше моделей OY, в моделях SKT часто используются не-

зависимые и зависимые источники тока и напряжения. Эти источники позволяют задавать уровень напряжения или тока в подсхеме устройства, причём существует гибкая система описания этих напряжений и токов.

НЕЗАВИСИМЫЕ ИДЕАЛЬНЫЕ ИСТОЧНИКИ НАПЯЖЕНИЯ И ТОКА

Форма описания источников:

```
V<Имя> <Узел(+)><Узел(-)>
[DC<Амплитуда> [фаза]] [<Сигнал> (<Параметры>)]
I<Имя> <Узел(+)><Узел(-)>
[AC<Амплитуда> [фаза]] [<Сигнал> (<Параметры>)]
```

Положительным направлением тока считается направление от узла (+) через источник к узлу (-). Для источников могут указываться значения, необходимые для расчётов по постоянному току и переходных процессов DC (по

умолчанию – 0), для частотного анализа AC (амплитуда по умолчанию – 0) фаза указывается в градусах. По умолчанию фаза равна нулю. Если задаётся значение сигнала для переходного процесса, то <Сигнал> может принимать значения: Exp – экспоненциальный источник, Pulse – импульсный источник, SFFM – частотно модулированный источник, Sin – синусоидальный источник [2].

Примеры:

- V2 3 0 DC 12 – источник напряжения 12 В, включенный между узлами 3 и 0;
- VSIN 2 0 SIN(0 0,2V 1MEG) – источник синусоидального напряжения 0,2 В с постоянной составляющей 0 В и частотой 1 МГц;
- I3 4 11 DC 2mA – источник тока 2 мА, включенный между узлами 4 и 11;
- ISIN 2 0 SIN(0 0,2m 1000) – источник синусоидального тока 0,2 мА с постоянной составляющей 0 мА и частотой 1000 Гц.

ЗАВИСИМЫЕ ИСТОЧНИКИ НАПЯЖЕНИЯ И ТОКА

Зависимые источники (см. рис. 8) широко используются при построении макромоделей и функциональных блоков. Их применение позволяет простыми средствами имитировать любые зависимости между напряжениями и токами. Кроме этого, с их помощью очень просто организовать передачу информации от одного функционального блока к другому.

В Altium Designer встроены следующие типы зависимых источников (которые расположены в библиотеке Simulation Sources.IntLib):

- E – источник напряжения, управляемый напряжением (ИНУН);
- F – источник тока, управляемый током (ИТУТ);
- G – источник тока, управляемый напряжением (ИТУН);




Рис. 6. АЧХ усилителя без обратной связи

```
*LM339
*Quad SnglSup pkg:DIP14 (A:5,4,3,12,2)(B:7,6,3,12,1)(C:9,8,3,12,14)(D:11,10,3,12,13)
* Connections:
* Non-Inverting Input
* Inverting Input
* Positive Power Supply
* Negative Power Supply
* Out
*
SUBCKT LM339 1 2 3 4 5
F1 9 3 V1 1
IEE 3 7 DC 100.0E-6
V11 21 1 DC .75
V12 22 2 DC .75
Q1 9 21 7 QIN
Q2 8 22 7 QIN
Q3 9 8 4 QMO
Q4 8 8 4 QMT
.MODEL QIN PNP(IS=800.0E-18 BF=2E3)
.MODEL QMI NPN(IS=800.0E-18 BF=1002)
.MODEL QMO NPN(IS=800.0E-18 BF=1000 CJC=1E-15 TR=807.4E-9)
E1 10 4 9 4 1
V1 10 11 DC 0
Q5 5 11 4 QOC
.MODEL QOC NPN(IS=800.0E-18 BF=20.29E3 CJC=1E-15 TF=942.6E-12 TR=543.8E-9)
DP 4 3 DX
RP 3 4 4E3E3
.MODEL DX D(IS=800.0E-18)
ENDS LM339
```

Рис. 7. Модель компаратора LM339

- Н – источник напряжения, управляемый током (ИНУТ).

Форма описания включения в схему: Первый символ<Имя> <Узел(+)><Узел(-)> <Передачная функция>.

Первый символ имени должен соответствовать типу источника. Положительным направлением тока считается направление от узла (+) через источник к узлу (-). Далее указывается передаточная функция, которая может быть описана разными способами:

- степенным полиномом: Poly (<Выражение>);
- формулой: Value=(<Выражение>);
- таблицей: Table (<Выражение>);
- преобразованием Лапласа: Laplace (<Выражение>);
- частотной таблицей: Freq (<Выражение>);
- полиномом Чебышева: Chebyshev (<Выражение>).

Источники сигналов сложной формы и разнообразные нелинейные приборы удобно моделировать с помощью источников напряжения (E) или тока (G), т.к. они позволяют использовать параметры и любые функции от узловых потенциалов, токов (через независимые источники напряжения) и времени в соответствии со следующей формой описания включения в схему:

```
E<Имя> <+Узел> <-Узел>
Value={<Выражение>} G<Имя>
<+Узел> <-Узел> Value ={<Выражение>}
```

После ключевого слова *Value* в фигурных скобках приводится алгебраическое выражение, зависящее от узловых потенциалов, разности узловых потенциалов, токов через независимые источники напряжения и времени.

Создадим зависимый источник сложной формы: ESIGNAL 2 0 VALUE = {50mV* SIN(6.28*100kHz*Time)*V(3,4)} GPW 6 0 + Value = {V(5)*I(VP)}. Здесь для текущего времени выделено ключевое слово *Time*.

Область управляемых источников с нелинейным законом управления весьма обширна. С их помощью легко имитировать нелинейные радиоэлементы со сложной ВАХ. В частности, нелинейные проводимости или резисторы имитируются в программе *Pspice* с помощью источников напряжения, управляемых собственным током, или источников тока, управляемых собственным напряжением. Пример имитации нелинейного резистора (см. рис. 5) и его описание имеют вид:




Рис. 8. Зависимые идеальные источники

```
G1 1 2 Value={F(V(G))}.
```

Нелинейная функция $F(VG)$, описывающая нелинейную зависимость тока от падения напряжения на резисторе, должна быть определена в задании на моделирование до строки с описанием источника G , например:

```
Func F(X)=1e-6.(1-Exp(-40.X))
```

В режимах *Tran* и *DC* значения источников $E<Имя>$, $G<Имя>$ вычисляются согласно приведённым в фигурных скобках выражениям. Если это выражение представляет собой линейную функцию нескольких переменных, то в режиме *AC* данный источник представляет собой линейный управляемый источник. При этом, если в выражение входит переменная *Time*, она полагается равной нулю. Если же это выражение представляет собой нелинейную функцию одной переменной, то после расчёта режима цепи по постоянному току выражение в фигурных скобках линеаризуется, и в частотной области такой источник представляет собой линеаризованный управляемый источник. Нелинейную функцию нескольких переменных при расчёте частотных характеристик использовать нельзя – результаты будут непредсказуемы. Эти же замечания справедливы и для описываемого ниже табличного задания управляемых источников.

Удобно создавать модели нелинейных приборов с помощью табличного задания передаточной функции. Здесь входом таблицы является <Выражение>, которое содержит любую комбинацию напряжений и токов:

```
E<Имя> <+Узел> <-Узел> Table
{<Выражение>} <<Аргумент> <Функция>
G<Имя> <+Узел> <-Узел> Table
{<Выражение>} <<Аргумент> <Функция>
```

При обращении к управляемому источнику вычисляется значение выражения и берётся значение функции из таблицы с помощью линейной интер-

поляции между опорными точками, заданными парами чисел (<Аргумент>, <Функция>). Например, ВАХ туннельного диода, рассматриваемого как нелинейная проводимость (см. рис. 5), может быть задана в виде:

```
GR 1 2 TABLE (V(GR)) = (0,0)
(.01,1mA) (.02,1.1mA) (0.05,2mA)
(.06,3mA) (.065,3.2mA)
+ (.8,1.5mA) (1,3mA) (1.5,5mA)
```

Координаты опорных точек задаются в порядке возрастания аргумента. Заметим, что отсутствие аппроксимации табличных значений сплайнами более высоких порядков, чем линейная интерполяция, в ряде случаев приводит к большим погрешностям и проблемам со сходимостью.

С помощью управляемых источников, передаточная функция которых задаётся с помощью преобразования Лапласа, можно описывать свойства приборов в частотной области:

```
E<имья> <-узел><-узел> Laplace
{<выражение>}=[<передаточная функция в S-области>]
G<имья><+узел><-узел> Laplace
{<выражение>}=[<передаточная функция в S-области>]
```

По директиве *AC* определяются значения комплексного коэффициента передачи блока на каждой частоте. При расчёте рабочей точки по постоянному току и по директиве *DC*, берётся значение коэффициента передачи на нулевой частоте (поэтому изображения по Лапласу всех узловых потенциалов не должны иметь составляющих типа $1/s$). По директиве *Tran* выходная переменная блока вычисляется как интеграл Дюамеля (свёртка входного воздействия с импульсной характеристикой блока), что значительно увеличивает длительность расчётов.

Возможны следующие варианты задания линейных блоков. Передаточную функцию активного RC-фильтра можно задать следующим образом:

```
EARC 2 0 Laplace {V(9)}={5/(1+0.01*s)}
```




Рис. 9. Использование моделей в формате *.LB

Для описания фильтров можно использовать зависимые источники с чебышевской передаточной функцией:

```
E<Имя> <+Узел> <-Узел> Chebyshev
{<Выражение>}=<-Тип> <Граничная
частота> +<Затухание>
G<Имя> <+Узел> <-Узел> Chebyshev
{<Выражение>}=<-Тип> <Граничная
частота> +<Затухание>
```

Здесь <Type> – тип фильтра, принимающий значения: LP – фильтр нижних частот, HP – фильтр верхних частот, BP – полосовой фильтр, BR – режекторный фильтр.

Для получения желаемой характеристики фильтра задаётся список граничных частот (для ФНЧ и ФВЧ задаются две частоты, для ПФ и РФ – четыре), а затем список затуханий на этих частотах в децибелах. При этом порядок следования граничных частот может быть любым. Такие фильтры описываются следующим образом:

```
E1 3 4 Chebyshev {V(10)}=LP 800
1.2K 0.1 dB 50dB
E2 5 6 Chebyshev {V(10)}=BP 800
1.2K 2K 3K 0.1dB 50dB
```

Для описания свойств приборов в частотной области можно использовать зависимые источники с табличным описанием комплексной передаточной функции, задаваемой по формату:

```
E<Имя> <+Узел> <-Узел> Freq {<Выражение>} [Keyword]
+ <<Частота>, <Модуль>, <Фаза>>
[Delay]=<Задержка>
G<Имя> <+Узел><-Узел> Freq {<Выражение>} [Keyword]
```

```
+ <<Частота>, <Модуль>, <Фаза>>
[Delay]=<Задержка>
```

Здесь частота задаётся в герцах, модуль передаточной функции – в децибелах, фаза передаточной функции – в градусах. Значения опорных точек указываются в порядке возрастания частоты. Максимальное количество точек 2048. Например, передаточная функция типа фильтра верхних частот задаётся следующим образом:

```
EHighpass 2 0 FREQ {V(9)} (0, -
60, 69.1) (2kHz, -3, 45) (5kHz, 0, 0)
```

С помощью ключевого слова *Keyword* изменяется способ задания таблицы передаточной функции:

- *MAG* – задание абсолютных значений передаточной функции;
- *DB* – задание значений передаточной функции в децибелах (принимается по умолчанию);
- *RAD* – задание фазы в радианах;
- *DEG* – задание фазы в градусах (принимается по умолчанию);
- *R_I* – задание действительной и мнимой части передаточной функции вместо её модуля и фазы.

Ключевое слово *Delay* задаёт дополнительную задержку, которая принимается во внимание при расчёте фазовой характеристики фильтра.

БИБЛИОТЕКИ МОДЕЛЕЙ И МОДЕЛИ МИКРОСХЕМ

Модели в форматах *MDL* и *CKT* могут храниться как по отдельности, по одной в каждом файле, так и группами, в одном файле. Например, для формирования набора моделей цифровых микросхем можно использовать файл *DIG-*

ITAL.LB, внутри которого содержится описание нескольких микросхем. Начало описания определяется меткой *.SUBCKT NAME* (или *.MODEL NAME*) и заканчивается командой *.ENDS NAME*. При использовании файла с расширением **.LB*, его также требуется поместить в структуру проекта, в котором находится библиотека с компонентами. Далее выполняются те же действия, что и при подключении одиночной модели, но теперь при выборе файла **.LB* будет доступен перечень моделей, описание которых содержится в файле (см. рис. 9).

Следует обратить внимание, что некоторые поставщики радиокомпонентов поставляют модели в форматах, отличных от перечисленных выше. В этом случае желательно просмотреть модель в обычном текстовом редакторе, и, при наличии в нем команд *.SUBCKT* или *.MDL*, можно изменить расширение файла на **.LB* и использовать по описанной выше методике.

Для описания цифровых микросхем применяется более сложный, специализированный язык системы моделирования *SimCode*. Модель цифровой микросхемы представляет собой описание входов и выходов микросхемы, их взаимосвязей, алгоритма функционирования микросхемы и её электрических характеристик (задержки сигналов, скорость переключения и т.д.) [4]. Язык *SimCode* позволяет при разработке моделей задавать статические и динамические характеристики микросхем, учитывать влияние температуры, напряжения питания и нагрузочную способность ИС. Подробное описание языка *SimCode*, рекомендации по разработке моделей цифровых микросхем и пример модели в исходном тексте приведены в электронной документации *Altium: TR0117 Digital SimCode Reference.pdf*, *AP0139 Creating and Linking a Digital SimCode Model.pdf*.

ЛИТЕРАТУРА

1. Кеун Дж. OrCAD Pspice. Анализ электрических цепей. Питер, 2008.
2. Сабунин А.Е. Altium Designer. Новые решения в проектировании электронных устройств. Солон-Пресс, 2009.
3. Разевиг В.Д. Схемотехническое моделирование с помощью Micro-Cap 7. Горячая линия – Телеком, 2003.
4. Татаринцев В.Д. К вопросу разработки моделей цифровых микросхем. EDA Express № 15, 2007.
5. Петраков О. Создание аналоговых PSPICE-моделей радиоэлементов. Схемотехника № 2-12. 2001; № 1-3, 2002.